The Wallace–Simson Line Theorem in Isabelle/HOL

Arthur Freitas Ramos David Barros Hulak
Ruy J. G. B. de Queiroz

August 5, 2026

Abstract

We formalize the Wallace–Simson line theorem in Isabelle/HOL. Let ABC be a nondegenerate triangle and let M be a point on its circumcircle. Dropping the perpendiculars from M to the three side lines BC, CA, AB gives feet P, Q, R; the theorem asserts that P, Q, R are collinear. The line through them is the *Simson line*, also called the *Wallace line* after William Wallace, whose published work on geometrical porisms is the standard early source for this result. We also prove the converse: if the three feet are collinear, then M lies on the circumcircle of ABC.

The development uses complex coordinates.

Contents

1 Introduction

The Wallace–Simson line is a classical incidence result about a triangle and a point on its circumcircle [1, 2, 4]. Given a nondegenerate triangle ABC and a point M on its circumcircle, the feet of the three perpendiculars from M to the side lines BC, CA, AB are collinear. The line so obtained is usually named after Robert Simson, but it is due to William Wallace [4]. The converse also holds and characterises the circumcircle: the feet are collinear *only* when the point lies on the circumcircle.

Our formalization works in the complex plane. It builds on the published *Complex Geometry* development [3] for its basic complex-plane notions; reducing the incidence statement to complex algebra then turns both directions of the theorem into short computations.

2 Complex-coordinate setup

We identify the Euclidean plane with $\mathbb{C}$. A point z lies on the circle of centre c and radius R when $|z - c| = R$, and a point z lies on the line through a and b when $z = a + t\,(b - a)$ for some real t. Three points a, b, c are collinear when

$$\mathrm{Im}\big(\overline{(b-a)}\,(c-a)\big) = 0,$$

a real multiple of the planar cross product, and two displacement vectors u, v are perpendicular when $\mathrm{Re}(\overline{u}\,v) = 0$, their Euclidean inner product. A basic lemma records that collinearity is equivalent to the ratio $(c-a)/(b-a)$ being real whenever the three points are distinct enough for the ratio to be defined, and this real-ratio form is what the main proofs ultimately use.

3 The foot of a perpendicular

The foot of the perpendicular from m to the line through a and b is

$$\operatorname{foot}(a,b,m) = \frac{a+m}{2} + \frac{(b-a)\,\overline{(m-a)}}{2\,\overline{(b-a)}}.$$

We verify that this closed form is meaningful: $\operatorname{foot}(a,b,m)$ lies on the line ab, and the residual $m - \operatorname{foot}(a,b,m)$ is perpendicular to the direction $b-a$. These properties also uniquely characterise the point on every nondegenerate line. Thus the three points named P, Q, R in the statement really are the feet of the perpendiculars onto the side lines, and the theorem is a statement about that geometric configuration rather than about an unmotivated algebraic expression.

4 The cross-ratio criterion

Four complex numbers w, x, y, z have cross ratio

$$(w,x;y,z) = \frac{(w-y)\,(x-z)}{(w-z)\,(x-y)}.$$

For four suitably distinct points, being concyclic is equivalent to this cross ratio being real. We need the forward direction: if w, x, y, z lie on a circle of centre o and radius $R>0$, then each point satisfies $\overline{z-o} = R^2/(z-o)$; substituting this into the conjugate of the cross ratio, every factor R^2 and every factor $(\cdot - o)$ cancels, leaving the cross ratio itself, so it equals its own conjugate and is real.

5 The forward theorem

Assume M lies on the circumcircle of ABC. Working relative to the centre o, each pairwise difference of feet on the circle collapses to a clean product; concretely

$$\operatorname{foot}(b,c,m) - \operatorname{foot}(a,b,m) = \frac{(c-a)\,(m-b)}{2\,(m-o)}, \qquad \operatorname{foot}(c,a,m) - \operatorname{foot}(a,b,m) = \frac{(c-b)\,(m-a)}{2\,(m-o)}.$$

Their ratio is

$$\frac{\operatorname{foot}(b,c,m) - \operatorname{foot}(a,b,m)}{\operatorname{foot}(c,a,m) - \operatorname{foot}(a,b,m)} = \frac{(c-a)\,(m-b)}{(c-b)\,(m-a)} = (c,m;a,b),$$

the cross ratio of the four concyclic points c, m, a, b. By the cross-ratio criterion this ratio is real, and reality of the ratio of the two foot differences is exactly the collinearity of the three feet. (The degenerate case $m=a$, where two of the feet coincide, is handled separately.)

6 The converse

For the converse we drop the assumption that M lies on the circle and assume instead that the three feet are collinear. Using the same closed forms for the foot differences — now keeping the conjugate substitutions symbolic — the collinearity relation $\mathrm{Im}\big(\overline{(P-R)}\,(Q-R)\big) = 0$ becomes, after clearing the three circle denominators $(a-o)$, $(b-o)$, $(c-o)$, a single algebraic identity whose right-hand side factors as

$$R^4\,(a-b)\,(a-c)\,(b-c)\,\big(R^2 - \overline{(m-o)}\,(m-o)\big).$$

The three vertex differences are nonzero because ABC is nondegenerate, and the vanishing of the left-hand side forces $\overline{(m-o)}\,(m-o) = R^2$, that is $|m-o| = R$. Hence M lies on the circumcircle.

7 Formalization notes

The development uses the published `Complex_Geometry` session for its basic collinearity infrastructure. The forward theorem `simson_line` takes a point on the circumcircle and concludes collinearity of the three feet; the converse `simson_line_converse` takes collinearity of the feet and concludes that the point lies on the circumcircle. The theorem `simson_line_iff` combines them into the corresponding characterisation. The only structural hypothesis is that A, B, C are not collinear, together with a positive radius; the distinctness of the vertices is derived from non-collinearity. The closed-form foot is shown to be the unique point on the side line whose displacement from M is perpendicular to that line, so the formal statements precisely capture the Wallace–Simson line theorem and its converse.

```
theory Simson-Complex-Geometry
  imports Complex-Geometry.Elementary-Complex-Geometry
begin

lemma Im-zero-self-cnj:
  assumes Im W = 0 shows W = cnj W
  using assms by (simp add: complex-eq-iff)

lemma im-zero-cross:
  fixes A B :: complex
  assumes h: Im (cnj A * B) = 0
  shows cnj A * B = A * cnj B
proof -
  have cnj A * B = cnj (cnj A * B) by (rule Im-zero-self-cnj[OF h])
  also have ... = A * cnj B by (simp add: complex-cnj-mult)
  finally show ?thesis .
```

qed

lemma *collinear-iff-cross*:
  collinear a b c $\longleftrightarrow$ *Im* (*cnj* ($b - a$) $*$ ($c - a$)) = *0*
proof (*cases* $a = b$)
  case *True*
  then show *?thesis* **by** (*simp add*: *collinear-def*)
next
  case *False*
  have *hab*: $b - a \neq 0$ **using** *False* **by** *simp*
  have *ratio-real*:
    is-real (($c - a$) / ($b - a$)) $\longleftrightarrow$
      ($c - a$) $*$ *cnj* ($b - a$) = ($b - a$) $*$ *cnj* ($c - a$)
    using *is-real-div*[*of* $b - a$ $c - a$] *hab* **by** *simp*
  have *cross-real*:
    is-real (*cnj* ($b - a$) $*$ ($c - a$)) $\longleftrightarrow$
      cnj ($b - a$) $*$ ($c - a$) = ($b - a$) $*$ *cnj* ($c - a$)
  proof
    assume *h*: *is-real* (*cnj* ($b - a$) $*$ ($c - a$))
    then show *cnj* ($b - a$) $*$ ($c - a$) = ($b - a$) $*$ *cnj* ($c - a$)
      by (*rule im-zero-cross*)
  next
    assume *h*: *cnj* ($b - a$) $*$ ($c - a$) = ($b - a$) $*$ *cnj* ($c - a$)
    have *cnj* (*cnj* ($b - a$) $*$ ($c - a$)) = *cnj* ($b - a$) $*$ ($c - a$)
      using *h* **by** (*simp add*: *complex-cnj-mult*)
    then show *is-real* (*cnj* ($b - a$) $*$ ($c - a$))
      by (*metis Reals-cnj-iff complex-is-Real-iff*)
  qed
  show *?thesis*
    unfolding *collinear-def*
    using *False ratio-real cross-real* **by** (*simp add*: *mult.commute*)
qed

8 Circles, lines, and the ordinary complex cross ratio

A point z lies on the circle of centre c and radius R when its distance to the centre equals R.

definition *oncircle* :: *complex* $\Rightarrow$ *real* $\Rightarrow$ *complex* $\Rightarrow$ *bool*
  where *oncircle c R z* $\longleftrightarrow$ *cmod* ($z - c$) = R

A point z lies on the line through a and b when it is an affine combination of them with a real parameter.

definition *on-line* :: *complex* $\Rightarrow$ *complex* $\Rightarrow$ *complex* $\Rightarrow$ *bool*
  where *on-line a b z* $\longleftrightarrow$ ($\exists$ *t*::*real*. $z = a +$ *of-real* $t * (b - a)$)

Four points are concyclic when they share a common circle of positive radius; three points are concyclic when they share such a circle.

definition *concyclic* :: *complex* ⇒ *complex* ⇒ *complex* ⇒ *complex* ⇒ *bool*
  where *concyclic a b c d* ⟷
    (∃ *ctr R*. *0* < *R* ∧ *oncircle ctr R a* ∧ *oncircle ctr R b* ∧ *oncircle ctr R c* ∧ *oncircle ctr R d*)

definition *concyclic3* :: *complex* ⇒ *complex* ⇒ *complex* ⇒ *bool*
  where *concyclic3 a b c* ⟷
    (∃ *ctr R*. *0* < *R* ∧ *oncircle ctr R a* ∧ *oncircle ctr R b* ∧ *oncircle ctr R c*)

The ordinary cross ratio of four complex numbers.

definition *complex-cross-ratio* :: *complex* ⇒ *complex* ⇒ *complex* ⇒ *complex* ⇒ *complex*
  where *complex-cross-ratio w x y z* = (*w* − *y*) ∗ (*x* − *z*) / ((*w* − *z*) ∗ (*x* − *y*))

9 Elementary facts about circles

lemma *oncircle-norm-sq*:
  assumes *oncircle c R z*
  shows (*z* − *c*) ∗ *cnj* (*z* − *c*) = *complex-of-real* (R^2)
proof −
  have (*z* − *c*) ∗ *cnj* (*z* − *c*) = *complex-of-real* ((*cmod* (*z* − *c*))^2)
    by (*metis complex-norm-square*)
  then show *?thesis* **using** *assms* **by** (*simp add*: *oncircle-def*)
qed

lemma *oncircle-ne-centre*:
  assumes *oncircle c R z* **and** *0* < *R*
  shows *z* ≠ *c*
proof
  assume *z* = *c*
  then have *cmod* (*z* − *c*) = *0* **by** *simp*
  with *assms* **show** *False* **by** (*simp add*: *oncircle-def*)
qed

lemma *cnj-diff-on-circle*:
  assumes *oncircle c R z* **and** *0* < *R*
  shows *cnj* (*z* − *c*) = *complex-of-real* (R^2) / (*z* − *c*)
proof −
  have *z* − *c* ≠ *0* **using** *oncircle-ne-centre*[*OF assms*] **by** *simp*
  moreover have (*z* − *c*) ∗ *cnj* (*z* − *c*) = *complex-of-real* (R^2)
    using *oncircle-norm-sq*[*OF assms*(*1*)] **.**
  ultimately show *?thesis*
    by (*metis nonzero-mult-div-cancel-left*)
qed

10 The cross-ratio criterion

The algebraic heart of the criterion: replacing each of four points by the reciprocal expression that describes a point on a circle leaves the cross ratio unchanged.

lemma *complex-cross-ratio-reciprocal-aux*:
  fixes $A\ B\ C\ D\ k$ :: *complex*
  assumes $A \neq 0$ **and** $B \neq 0$ **and** $C \neq 0$ **and** $D \neq 0$
    and $A \neq D$ **and** $B \neq C$ **and** $k \neq 0$
  shows $(k/A - k/C) * (k/B - k/D) / ((k/A - k/D) * (k/B - k/C))$
      $= (A - C) * (B - D) / ((A - D) * (B - C))$
  using *assms* **by** (*simp add*: *divide-simps*) *algebra*

If four points lie on a common circle of positive radius (and the two relevant differences are nonzero), then their cross ratio is real.

lemma *concyclic-complex-cross-ratio-real*:
  assumes *a*: *oncircle ctr R a* **and** *b*: *oncircle ctr R b*
    and *c*: *oncircle ctr R c* **and** *d*: *oncircle ctr R d*
    and *pos*: $0 < R$ **and** *ad*: $a \neq d$ **and** *bc*: $b \neq c$
  shows *complex-cross-ratio a b c d* $\in \mathbb{R}$
proof −
  define A **where** $A = a - ctr$
  define B **where** $B = b - ctr$
  define C **where** $C = c - ctr$
  define D **where** $D = d - ctr$
  define k **where** $k =$ *complex-of-real* (R^2)
  have *kne*: $k \neq 0$ **using** *pos* **by** (*simp add*: *k-def*)
  have *Ane*: $A \neq 0$ **using** *oncircle-ne-centre*[*OF a pos*] **by** (*simp add*: *A-def*)
  have *Bne*: $B \neq 0$ **using** *oncircle-ne-centre*[*OF b pos*] **by** (*simp add*: *B-def*)
  have *Cne*: $C \neq 0$ **using** *oncircle-ne-centre*[*OF c pos*] **by** (*simp add*: *C-def*)
  have *Dne*: $D \neq 0$ **using** *oncircle-ne-centre*[*OF d pos*] **by** (*simp add*: *D-def*)
  have *ADne*: $A \neq D$ **using** *ad* **by** (*simp add*: *A-def D-def*)
  have *BCne*: $B \neq C$ **using** *bc* **by** (*simp add*: *B-def C-def*)
  — conjugates of the four points on the circle
  have *cA*: *cnj* $(a - ctr) = k\ /\ A$
    using *cnj-diff-on-circle*[*OF a pos*] **by** (*simp add*: *A-def k-def*)
  have *cB*: *cnj* $(b - ctr) = k\ /\ B$
    using *cnj-diff-on-circle*[*OF b pos*] **by** (*simp add*: *B-def k-def*)
  have *cC*: *cnj* $(c - ctr) = k\ /\ C$
    using *cnj-diff-on-circle*[*OF c pos*] **by** (*simp add*: *C-def k-def*)
  have *cD*: *cnj* $(d - ctr) = k\ /\ D$
    using *cnj-diff-on-circle*[*OF d pos*] **by** (*simp add*: *D-def k-def*)
  — conjugates of the six differences appearing in the cross ratio
  have *e-ac*: *cnj* $(a - c) = k/A - k/C$
    using *cA cC* **by** (*metis complex-cnj-diff diff-add-cancel diff-diff-eq2*)
  have *e-bd*: *cnj* $(b - d) = k/B - k/D$
    using *cB cD* **by** (*metis complex-cnj-diff diff-add-cancel diff-diff-eq2*)
  have *e-ad*: *cnj* $(a - d) = k/A - k/D$
    using *cA cD* **by** (*metis complex-cnj-diff diff-add-cancel diff-diff-eq2*)

have *e-bc*: *cnj* (*b* − *c*) = *k*/*B* − *k*/*C*
using *cB cC* **by** (*metis complex-cnj-diff diff-add-cancel diff-diff-eq2*)
have *cnj* (*complex-cross-ratio a b c d*)
= *cnj* (*a* − *c*) ∗ *cnj* (*b* − *d*) / (*cnj* (*a* − *d*) ∗ *cnj* (*b* − *c*))
by (*simp only*: *complex-cross-ratio-def complex-cnj-divide complex-cnj-mult*)
also have . . . = (*k*/*A* − *k*/*C*) ∗ (*k*/*B* − *k*/*D*) / ((*k*/*A* − *k*/*D*) ∗ (*k*/*B* − *k*/*C*))
by (*simp only*: *e-ac e-bd e-ad e-bc*)
also have . . . = (*A* − *C*) ∗ (*B* − *D*) / ((*A* − *D*) ∗ (*B* − *C*))
using *complex-cross-ratio-reciprocal-aux*[*OF Ane Bne Cne Dne ADne BCne kne*] **.**
also have . . . = *complex-cross-ratio a b c d*
by (*simp add*: *complex-cross-ratio-def A-def B-def C-def D-def*)
finally have *cnj* (*complex-cross-ratio a b c d*) = *complex-cross-ratio a b c d* **.**
thus *?thesis* **by** (*simp add*: *Reals-cnj-iff*)
qed

If three points lie on a circle and the cross ratio of a fourth point with them is real, the fourth point lies on the same circle.

lemma *complex-cross-ratio-real-on-circle*:
assumes *hc*: *oncircle ctr R c* **and** *hp*: *oncircle ctr R p* **and** *hq*: *oncircle ctr R q*
and *pos*: *0* < *R*
and *cp*: *c* ≠ *p* **and** *cq*: *c* ≠ *q* **and** *pq*: *p* ≠ *q* **and** *mq*: *m* ≠ *q*
and *crreal*: *complex-cross-ratio m c p q* ∈ ℝ
shows *oncircle ctr R m*
proof −
define *k* **where** *k* = *complex-of-real* (R^2)
define *G* **where** *G* = *c* − *ctr*
define *P* **where** *P* = *p* − *ctr*
define *K* **where** *K* = *q* − *ctr*
define *M* **where** *M* = *m* − *ctr*
define *cM* **where** *cM* = *cnj* (*m* − *ctr*)
have *kne*: *k* ≠ *0* **using** *pos* **by** (*simp add*: *k-def*)
have *Gne*: *G* ≠ *0* **using** *oncircle-ne-centre*[*OF hc pos*] **by** (*simp add*: *G-def*)
have *Pne*: *P* ≠ *0* **using** *oncircle-ne-centre*[*OF hp pos*] **by** (*simp add*: *P-def*)
have *Kne*: *K* ≠ *0* **using** *oncircle-ne-centre*[*OF hq pos*] **by** (*simp add*: *K-def*)
have *GK*: *G* ≠ *K* **using** *cq* **by** (*simp add*: *G-def K-def*)
have *GP*: *G* ≠ *P* **using** *cp* **by** (*simp add*: *G-def P-def*)
have *KP*: *K* ≠ *P* **using** *pq* **by** (*simp add*: *K-def P-def*)
have *MK*: *M* ≠ *K* **using** *mq* **by** (*simp add*: *M-def K-def*)
— conjugates of the three circle points
have *cG*: *cnj* (*c* − *ctr*) = *k* / *G*
using *cnj-diff-on-circle*[*OF hc pos*] **by** (*simp add*: *G-def k-def*)
have *cP*: *cnj* (*p* − *ctr*) = *k* / *P*
using *cnj-diff-on-circle*[*OF hp pos*] **by** (*simp add*: *P-def k-def*)
have *cK*: *cnj* (*q* − *ctr*) = *k* / *K*
using *cnj-diff-on-circle*[*OF hq pos*] **by** (*simp add*: *K-def k-def*)
— conjugates of the four differences of the cross ratio
have *e-mp*: *cnj* (*m* − *p*) = *cM* − *k* / *P*
using *cP* **unfolding** *cM-def* **by** (*metis complex-cnj-diff diff-add-cancel diff-diff-eq2*)

have *e-mq*: *cnj* $(m - q) = cM - k\ /\ K$
using *cK* **unfolding** *cM-def* **by** (*metis complex-cnj-diff diff-add-cancel diff-diff-eq2*)
have *e-cq*: *cnj* $(c - q) = k\ /\ G - k\ /\ K$
using *cG cK* **by** (*metis complex-cnj-diff diff-add-cancel diff-diff-eq2*)
have *e-cp*: *cnj* $(c - p) = k\ /\ G - k\ /\ P$
using *cG cP* **by** (*metis complex-cnj-diff diff-add-cancel diff-diff-eq2*)
— denominators are nonzero
have *KcMk*: $K * cM - k \neq 0$
proof −
have *cnj* $(m - q) = (K * cM - k)\ /\ K$
using *e-mq Kne* **by** (*simp add*: *field-simps*)
moreover have *cnj* $(m - q) \neq 0$ **using** *mq* **by** *simp*
ultimately show *?thesis* **using** *Kne* **by** *auto*
qed
— simplify the conjugate cross ratio
have *cnjcr*: *cnj* (*complex-cross-ratio m c p q*)
$= (P * cM - k) * (G - K)\ /\ ((K * cM - k) * (G - P))$
proof −
have *cnj* (*complex-cross-ratio m c p q*)
$=$ *cnj* $(m - p) *$ *cnj* $(c - q)\ /\ ($*cnj* $(m - q) *$ *cnj* $(c - p))$
by (*simp only*: *complex-cross-ratio-def complex-cnj-divide complex-cnj-mult*)
also have ... $= (cM - k/P) * (k/G - k/K)\ /\ ((cM - k/K) * (k/G - k/P))$
by (*simp only*: *e-mp e-mq e-cq e-cp*)
also have ... $= (P * cM - k) * (G - K)\ /\ ((K * cM - k) * (G - P))$
using *Gne Pne Kne kne* **by** (*simp add*: *divide-simps*) *algebra*
finally show *?thesis* **.**
qed
— the cross ratio itself in the shifted coordinates
have *cr*: *complex-cross-ratio m c p q* $= (M - P) * (G - K)\ /\ ((M - K) * (G$
$- P))$
by (*simp add*: *complex-cross-ratio-def M-def P-def K-def G-def*)
— reality of the cross ratio
from *crreal* **have** *cnj* (*complex-cross-ratio m c p q*) = *complex-cross-ratio m c p*
q
by (*simp add*: *Reals-cnj-iff*)
with *cnjcr cr*
have *eq1*: $(P * cM - k) * (G - K)\ /\ ((K * cM - k) * (G - P))$
$= (M - P) * (G - K)\ /\ ((M - K) * (G - P))$
by *simp*
— cancel the common nonzero factors and cross-multiply
have *heq*: $(P * cM - k) * (M - K) = (M - P) * (K * cM - k)$
proof −
have *Y*: $(K * cM - k) * (G - P) \neq 0$ **using** *KcMk GP* **by** *simp*
have *V*: $(M - K) * (G - P) \neq 0$ **using** *MK GP* **by** *simp*
have $(P * cM - k) * (G - K) * ((M - K) * (G - P))$
$= (M - P) * (G - K) * ((K * cM - k) * (G - P))$
using *eq1 Y V* **by** (*simp add*: *frac-eq-eq*)
then have $((P * cM - k) * (M - K)) * ((G - K) * (G - P))$
$= ((M - P) * (K * cM - k)) * ((G - K) * (G - P))$

by (*simp add*: *mult-ac*)
moreover have $(G - K) * (G - P) \neq 0$ **using** *GK GP* **by** *simp*
ultimately show *?thesis* **by** *simp*
qed
have $(K - P) * (M * cM - k) = 0$
using *heq* **by** (*simp add*: *algebra-simps*)
then have $M * cM = k$
using *KP* **by** (*metis mult-eq-0-iff right-minus-eq*)
— translate back to a distance statement
have $M * cM = \textit{complex-of-real}\ ((\textit{cmod}\ M)^2)$
unfolding *cM-def M-def* **by** (*metis complex-norm-square*)
with ‹$M * cM = k$› **have** $\textit{complex-of-real}\ ((\textit{cmod}\ M)^2) = \textit{complex-of-real}\ (R^2)$
by (*simp add*: *k-def*)
then have $(\textit{cmod}\ M)^2 = R^2$ **by** (*metis of-real-eq-iff*)
then have $\textit{cmod}\ M = R$ **using** *pos* **by** (*simp add*: *power2-eq-imp-eq*)
thus *?thesis* **by** (*simp add*: *oncircle-def M-def*)
qed

11 Collinearity gives real ratios

lemma *on-line-ratio-real*:
assumes *on-line a b z* **and** $b \neq z$
shows $(a - z) / (b - z) \in \mathbb{R}$
proof −
from *assms*(*1*) **obtain** t **where** z: $z = a + \textit{of-real}\ t * (b - a)$
by (*auto simp*: *on-line-def*)
have *az*: $a - z = \textit{of-real}\ (- t) * (b - a)$
using z **by** (*simp add*: *algebra-simps of-real-minus*)
have *bz*: $b - z = \textit{of-real}\ (1 - t) * (b - a)$
using z **by** (*simp add*: *algebra-simps of-real-diff*)
have *bane*: $b - a \neq 0$
proof
assume $b - a = 0$
with *bz* **have** $b - z = 0$ **by** *simp*
with *assms*(*2*) **show** *False* **by** *simp*
qed
have *tne*: $1 - t \neq 0$
proof
assume $1 - t = 0$
with *bz* **have** $b - z = 0$ **by** *simp*
with *assms*(*2*) **show** *False* **by** *simp*
qed
have $(a - z) / (b - z) = \textit{of-real}\ (- t) / \textit{of-real}\ (1 - t)$
using *az bz bane* **by** *simp*
also have $\ldots = \textit{of-real}\ (- t / (1 - t))$
by (*simp add*: *of-real-divide*)
finally show *?thesis* **by** *simp*
qed

end

theory *Simson*
  imports *Simson-Complex-Geometry*
begin

12 Perpendicularity and feet of perpendiculars

A point z lies on the circle of centre c and radius R when its distance to the centre equals R.

Two displacement vectors are perpendicular when the real part of $cnj\ u * v$ (their Euclidean inner product) vanishes.

definition *complex-perpendicular* :: *complex* ⇒ *complex* ⇒ *bool*
  where *complex-perpendicular u v* ⟷ *Re* (*cnj u* ∗ *v*) = *0*

The foot of the perpendicular from m to the line through a and b, in closed complex form.

definition *foot* :: *complex* ⇒ *complex* ⇒ *complex* ⇒ *complex*
  where *foot a b m* = (*a* + *m*)/*2* + (*b* − *a*) ∗ *cnj* (*m* − *a*) / (*2* ∗ *cnj* (*b* − *a*))

13 Collinearity as a real ratio

Three distinct-enough points are collinear exactly when the ratio of the two displacement vectors is real.

lemma *collinear-iff-real-ratio*:
  assumes *a* ≠ *c* **shows** *collinear a b c* = ((*b* − *a*)/(*c* − *a*) ∈ ℝ)
proof −
  have *hc*: *c* − *a* ≠ *0* **using** *assms* **by** *simp*
  have *E*: ((*b* − *a*)/(*c* − *a*) ∈ ℝ) = (*cnj* (*b* − *a*) ∗ (*c* − *a*) = (*b* − *a*) ∗ *cnj* (*c* − *a*))
    using *hc* **by** (*auto simp add*: *Reals-cnj-iff complex-cnj-divide field-simps*)
  have *collinear a b c* = (*Im* (*cnj* (*b* − *a*) ∗ (*c* − *a*)) = *0*)
    by (*simp add*: *collinear-iff-cross*)
  also have ... = (*cnj* (*b* − *a*) ∗ (*c* − *a*) = (*b* − *a*) ∗ *cnj* (*c* − *a*))
  proof
    assume *Im* (*cnj* (*b* − *a*) ∗ (*c* − *a*)) = *0*
    thus *cnj* (*b* − *a*) ∗ (*c* − *a*) = (*b* − *a*) ∗ *cnj* (*c* − *a*) **by** (*rule im-zero-cross*)
  next
    assume *cnj* (*b* − *a*) ∗ (*c* − *a*) = (*b* − *a*) ∗ *cnj* (*c* − *a*)
     hence *cnj* (*b* − *a*) ∗ (*c* − *a*) = *cnj* (*cnj* (*b* − *a*) ∗ (*c* − *a*)) **by** (*simp add*: *complex-cnj-mult*)
   thus *Im* (*cnj* (*b* − *a*) ∗ (*c* − *a*)) = *0* **by** (*metis Reals-cnj-iff complex-is-Real-iff*)
  qed
  finally show *?thesis* **using** *E* **by** *simp*
qed

Degenerate instances of collinearity, and its cyclic symmetry.

lemma *collinear-refl*: *collinear a b b*
by (*simp add*: *collinear-iff-cross algebra-simps*)

lemma *collinear-aba*: *collinear a b a*
by (*simp add*: *collinear-iff-cross algebra-simps*)

lemma *collinear-aac*: *collinear a a c*
by (*simp add*: *collinear-iff-cross algebra-simps*)

lemma *collinear-cycle*:
assumes *collinear a b c* **shows** *collinear b c a*
using *assms* **by** (*simp add*: *collinear-iff-cross algebra-simps*)

A non-collinear triple has three distinct vertices.

lemma *ncol-distinct*:
assumes ¬ *collinear a b c* **shows** $a \neq b \wedge a \neq c \wedge b \neq c$
using *assms* **by** (*metis collinear-aac collinear-aba collinear-refl*)

14 Points on a circle

On a circle, the squared distance to the centre is the constant R^2.

The converse: the norm identity characterises membership of a circle of positive radius.

lemma *oncircle-of-norm-sq*:
fixes *z ctr* :: *complex* **and** *R* :: *real*
assumes *h*: (*z* − *ctr*) ∗ *cnj* (*z* − *ctr*) = *complex-of-real* (*R*^*2*) **and** *pos*: *0* < *R*
shows *oncircle ctr R z*
proof −
have *complex-of-real* ((*cmod* (*z* − *ctr*))^*2*) = (*z* − *ctr*) ∗ *cnj* (*z* − *ctr*)
by (*rule complex-norm-square*)
also have ... = *complex-of-real* (*R*^*2*) **by** (*rule h*)
finally have *complex-of-real* ((*cmod* (*z* − *ctr*))^*2*) = *complex-of-real* (*R*^*2*) **.**
hence (*cmod* (*z* − *ctr*))^*2* = *R*^*2* **by** (*simp only*: *of-real-eq-iff*)
hence *cmod* (*z* − *ctr*) = *R* **using** *pos* **by** (*simp add*: *power2-eq-iff-nonneg*)
thus *?thesis* **by** (*simp add*: *oncircle-def*)
qed

15 The foot of a perpendicular

Two elementary conjugate identities.

lemma *of-real-Re-eq*: *complex-of-real* (*Re z*) = (*z* + *cnj z*)/*2*
by (*simp add*: *complex-add-cnj*)

lemma *Re-sub-cnj-half-zero*: *Re* ((*z* − *cnj z*)/*2*) = *0*
by (*simp add*: *complex-diff-cnj*)

The parameter of the foot along the line, before any circle assumption.

lemma *foot-ratio-gen*:
  fixes $a\ b\ m\ p\ q$ :: *complex*
  assumes $q \neq 0$ **and** $b - a \neq 0$
  shows $((a + m)/2 + (b - a)*p/(2*q) - a)/(b - a) = ((m - a)/(b - a) + p/q)/2$
  using *assms* **by** (*simp add*: *field-simps*)

A point lies on a line exactly when the displacement ratio is real.

lemma *on-line-iff*:
  assumes $a \neq b$ **shows** *on-line* $a\ b\ z = ((z - a)/(b - a) \in \mathbb{R})$
proof −
  have *hba*: $b - a \neq 0$ **using** *assms* **by** *simp*
  show *?thesis*
  proof
    assume *on-line* $a\ b\ z$
    then obtain t::*real* **where** $z = a +$ *of-real* $t * (b - a)$ **unfolding** *on-line-def* **by** *blast*
    hence $z - a =$ *of-real* $t * (b - a)$ **by** *simp*
    hence $(z - a)/(b - a) =$ *of-real* t **using** *hba* **by** (*simp add*: *field-simps*)
    thus $(z - a)/(b - a) \in \mathbb{R}$ **by** *simp*
  next
    assume $(z - a)/(b - a) \in \mathbb{R}$
    then obtain t::*real* **where** $(z - a)/(b - a) =$ *of-real* t **using** *Reals-cases* **by** *blast*
    hence $z - a =$ *of-real* $t * (b - a)$ **using** *hba* **by** (*simp add*: *field-simps*)
    hence $z = a +$ *of-real* $t * (b - a)$ **by** (*simp add*: *algebra-simps*)
    thus *on-line* $a\ b\ z$ **unfolding** *on-line-def* **by** *blast*
  qed
qed

The foot lies on its line.

lemma *foot-on-line*:
  assumes $a \neq b$ **shows** *on-line* $a\ b$ (*foot* $a\ b\ m$)
proof −
  have *hba*: $b - a \neq 0$ **using** *assms* **by** *simp*
  have *hcba*: *cnj* $(b - a) \neq 0$ **using** *assms* **by** *simp*
  have (*foot* $a\ b\ m - a)/(b - a) = ((m - a)/(b - a) +$ *cnj* $(m - a)/$*cnj* $(b - a))/2$
    unfolding *foot-def* **by** (*rule foot-ratio-gen*[*OF hcba hba*])
  also have ... $= ((m - a)/(b - a) +$ *cnj* $((m - a)/(b - a)))/2$
    by (*simp add*: *complex-cnj-divide*)
  also have ... = *complex-of-real* (*Re* $((m - a)/(b - a))$)
    by (*simp add*: *of-real-Re-eq*)
  finally have (*foot* $a\ b\ m - a)/(b - a) \in \mathbb{R}$ **by** *simp*
  thus *?thesis* **using** *on-line-iff*[*OF assms*] **by** *simp*
qed

The segment from m to its foot is perpendicular to the line.

lemma *foot-perp*:
assumes $a \neq b$ **shows** *complex-perpendicular* $(b - a)$ $(foot\ a\ b\ m - m)$
proof −
have *hba*: $cnj\ (b - a) \neq 0$ **using** *assms* **by** *simp*
have *eq*: $cnj\ (b - a) * (foot\ a\ b\ m - m) = ((b - a)*cnj\ (m - a) - cnj\ ((b - a)*cnj\ (m - a)))/2$
unfolding *foot-def* **using** *hba* **by** (*simp add*: *complex-cnj-mult field-simps*)
have $Re\ (cnj\ (b - a) * (foot\ a\ b\ m - m)) = 0$
unfolding *eq* **by** (*rule Re-sub-cnj-half-zero*)
thus *?thesis* **unfolding** *complex-perpendicular-def* **.**
qed

The preceding two properties uniquely characterise the perpendicular foot on every nondegenerate line.

lemma *foot-unique*:
assumes *ab*: $a \neq b$
and *zline*: *on-line* $a\ b\ z$
and *zperp*: *complex-perpendicular* $(b - a)$ $(z - m)$
shows $z = foot\ a\ b\ m$
proof −
obtain t :: *real* **where** *z*: $z = a + \textit{of-real}\ t * (b - a)$
using *zline* **unfolding** *on-line-def* **by** *blast*
have *fline*: *on-line* $a\ b\ (foot\ a\ b\ m)$
by (*rule foot-on-line*[*OF ab*])
then obtain s :: *real* **where** *f*: $foot\ a\ b\ m = a + \textit{of-real}\ s * (b - a)$
unfolding *on-line-def* **by** *blast*
have *fperp*: *complex-perpendicular* $(b - a)$ $(foot\ a\ b\ m - m)$
by (*rule foot-perp*[*OF ab*])
have *hzero*: $Re\ (cnj\ (b - a) * (z - foot\ a\ b\ m)) = 0$
using *zperp fperp* **unfolding** *complex-perpendicular-def*
by (*simp add*: *algebra-simps*)
have *hts*:
$(t - s) *$
$(Re\ (b - a) * Re\ (b - a) + Im\ (b - a) * Im\ (b - a)) = 0$
using *hzero* **unfolding** *z f*
by (*simp add*: *complex-mult-cnj algebra-simps*)
have *hnorm*:
$Re\ (b - a) * Re\ (b - a) + Im\ (b - a) * Im\ (b - a) \neq 0$
using *ab*
by (*auto simp*: *sum-squares-eq-zero-iff power2-eq-square complex-eq-iff*)
have $t - s = 0$
using *hts hnorm* **by** (*auto simp*: *mult-eq-0-iff*)
hence $t = s$ **by** *simp*
thus *?thesis* **unfolding** *z f* **by** *simp*
qed

16 Feet of the three perpendiculars from a point on the circle

Rewriting the foot when a and b lie on the circle: a pure field identity supporting the geometric rewrite that follows.

lemma *foot-general-gen*:
fixes *A B mu k* :: *complex*
assumes $A \neq 0$ **and** $B \neq 0$ **and** $A \neq B$ **and** $k \neq 0$
shows $(B - A)*(mu - k/A)/(2*(k/B - k/A)) = B*(k - mu*A)/(2*k)$
using *assms* **by** (*simp add*: *field-simps*)

Closed form for the foot when both line endpoints lie on the circle of centre *ctr* and radius *r*.

lemma *foot-general*:
fixes *a b m ctr* :: *complex* **and** *r* :: *real*
assumes *ha*: *oncircle ctr r a* **and** *hb*: *oncircle ctr r b* **and** *pos*: $0 < r$ **and** *dab*: $a \neq b$
shows *foot a b m* = *ctr* + (*a* − *ctr* + (*m* − *ctr*))/*2* + (*b* − *ctr*)∗(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)∗(*a* − *ctr*))/(*2*∗*complex-of-real* (*r*^*2*))
proof −
have *Ane*: *a* − *ctr* ≠ *0* **using** *oncircle-ne-centre*[*OF ha pos*] **by** *simp*
have *Bne*: *b* − *ctr* ≠ *0* **using** *oncircle-ne-centre*[*OF hb pos*] **by** *simp*
have *AB*: *a* − *ctr* ≠ *b* − *ctr* **using** *dab* **by** *simp*
have *R2ne*: *complex-of-real* (*r*^*2*) ≠ *0* **using** *pos* **by** *simp*
have *cnjA*: *cnj* (*a* − *ctr*) = *complex-of-real* (*r*^*2*)/(*a* − *ctr*) **using** *cnj-diff-on-circle*[*OF ha pos*] **.**
have *cnjB*: *cnj* (*b* − *ctr*) = *complex-of-real* (*r*^*2*)/(*b* − *ctr*) **using** *cnj-diff-on-circle*[*OF hb pos*] **.**
have *e1*: *cnj* (*m* − *a*) = *cnj* (*m* − *ctr*) − *complex-of-real* (*r*^*2*)/(*a* − *ctr*)
proof −
have *cnj* (*m* − *a*) = *cnj* (*m* − *ctr*) − *cnj* (*a* − *ctr*) **by** (*simp add*: *complex-cnj-diff*)
also have ... = *cnj* (*m* − *ctr*) − *complex-of-real* (*r*^*2*)/(*a* − *ctr*) **by** (*subst cnjA*) (*rule refl*)
finally show *?thesis* **.**
qed
have *e2*: *cnj* (*b* − *a*) = *complex-of-real* (*r*^*2*)/(*b* − *ctr*) − *complex-of-real* (*r*^*2*)/(*a* − *ctr*)
proof −
have *cnj* (*b* − *a*) = *cnj* (*b* − *ctr*) − *cnj* (*a* − *ctr*) **by** (*simp add*: *complex-cnj-diff*)
also have ... = *complex-of-real* (*r*^*2*)/(*b* − *ctr*) − *complex-of-real* (*r*^*2*)/(*a* − *ctr*) **by** (*simp only*: *cnjA cnjB*)
finally show *?thesis* **.**
qed
have *e3*: *b* − *a* = (*b* − *ctr*) − (*a* − *ctr*) **by** *simp*
have *g*:
((*b* − *ctr*) − (*a* − *ctr*)) ∗

(*cnj* (*m* − *ctr*) − *complex-of-real* (*r*^*2*) / (*a* − *ctr*)) /
(*2* ∗ (*complex-of-real* (*r*^*2*) / (*b* − *ctr*) −
complex-of-real (*r*^*2*) / (*a* − *ctr*)))
= (*b* − *ctr*) ∗
(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*) ∗ (*a* − *ctr*)) /
(*2* ∗ *complex-of-real* (*r*^*2*))
by (*rule foot-general-gen*[*OF Ane Bne AB R2ne*])
have *num*: (*b* − *a*)∗*cnj* (*m* − *a*) = ((*b* − *ctr*) − (*a* − *ctr*))∗(*cnj* (*m* − *ctr*) − *complex-of-real* (*r*^*2*)/(*a* − *ctr*))
by (*simp only*: *e3 e1*)
have *den*: *2*∗*cnj* (*b* − *a*) = *2*∗(*complex-of-real* (*r*^*2*)/(*b* − *ctr*) − *complex-of-real* (*r*^*2*)/(*a* − *ctr*))
by (*simp only*: *e2*)
have *key*: (*b* − *a*)∗*cnj* (*m* − *a*)/(*2*∗*cnj* (*b* − *a*)) = (*b* − *ctr*)∗(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)∗(*a* − *ctr*))/(*2*∗*complex-of-real* (*r*^*2*))
proof −
have
(*b* − *a*) ∗ *cnj* (*m* − *a*) / (*2* ∗ *cnj* (*b* − *a*))
= ((*b* − *ctr*) − (*a* − *ctr*)) ∗
(*cnj* (*m* − *ctr*) − *complex-of-real* (*r*^*2*) / (*a* − *ctr*)) /
(*2* ∗ (*complex-of-real* (*r*^*2*) / (*b* − *ctr*) −
complex-of-real (*r*^*2*) / (*a* − *ctr*)))
by (*simp only*: *num den*)
also have ... = (*b* − *ctr*)∗(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)∗(*a* − *ctr*))/(*2*∗*complex-of-real* (*r*^*2*))
by (*rule g*)
finally show *?thesis* **.**
qed
have *foot a b m* = (*a* + *m*)/*2* + (*b* − *a*)∗*cnj* (*m* − *a*)/(*2*∗*cnj* (*b* − *a*))
unfolding *foot-def* **by** *simp*
also have ... = (*a* + *m*)/*2* + (*b* − *ctr*)∗(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)∗(*a* − *ctr*))/(*2*∗*complex-of-real* (*r*^*2*))
using *key* **by** *simp*
also have ... = *ctr* + (*a* − *ctr* + (*m* − *ctr*))/*2* + (*b* − *ctr*)∗(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)∗(*a* − *ctr*))/(*2*∗*complex-of-real* (*r*^*2*))
by (*simp add*: *field-simps*)
finally show *?thesis* **.**
qed

The difference of the feet on BC and AB.

lemma *foot-diff-PR*:
fixes *a b c m ctr* :: *complex* **and** *r* :: *real*
assumes *ha*: *oncircle ctr r a* **and** *hb*: *oncircle ctr r b* **and** *hc*: *oncircle ctr r c*
and *pos*: *0* < *r* **and** *dab*: *a* ≠ *b* **and** *dbc*: *b* ≠ *c*
shows *foot b c m* − *foot a b m* = (*c* − *a*)∗(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)∗(*b* − *ctr*))/(*2*∗*complex-of-real* (*r*^*2*))
proof −
have *R2ne*: *complex-of-real* (*r*^*2*) ≠ *0* **using** *pos* **by** *simp*
have *fbc*: *foot b c m* = *ctr* + (*b* − *ctr* + (*m* − *ctr*))/*2* + (*c* − *ctr*)∗(*complex-of-real*

$(r^2) - cnj\ (m - ctr)*(b - ctr))/(2*complex\text{-}of\text{-}real\ (r^2))$
  by (*rule foot-general*[*OF hb hc pos dbc*])
have *fab*: *foot a b m* = *ctr* + (*a* − *ctr* + (*m* − *ctr*))/*2* + (*b* − *ctr*)*(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)*(*a* − *ctr*))/(*2**complex-of-real* (*r*^*2*))
  by (*rule foot-general*[*OF ha hb pos dab*])
show *?thesis* **unfolding** *fbc fab* **using** *R2ne* **by** (*simp add*: *field-simps*)
qed

The difference of the feet on CA and AB.

lemma *foot-diff-QR*:
fixes *a b c m ctr* :: *complex* **and** *r* :: *real*
assumes *ha*: *oncircle ctr r a* **and** *hb*: *oncircle ctr r b* **and** *hc*: *oncircle ctr r c*
  and *pos*: *0* < *r* **and** *dab*: *a* ≠ *b* **and** *dca*: *c* ≠ *a*
shows *foot c a m* − *foot a b m* = (*c* − *b*)*(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)*(*a* − *ctr*))/(*2**complex-of-real* (*r*^*2*))
proof −
have *R2ne*: *complex-of-real* (*r*^*2*) ≠ *0* **using** *pos* **by** *simp*
have *fca*: *foot c a m* = *ctr* + (*c* − *ctr* + (*m* − *ctr*))/*2* + (*a* − *ctr*)*(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)*(*c* − *ctr*))/(*2**complex-of-real* (*r*^*2*))
  by (*rule foot-general*[*OF hc ha pos dca*])
have *fab*: *foot a b m* = *ctr* + (*a* − *ctr* + (*m* − *ctr*))/*2* + (*b* − *ctr*)*(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)*(*a* − *ctr*))/(*2**complex-of-real* (*r*^*2*))
  by (*rule foot-general*[*OF ha hb pos dab*])
show *?thesis* **unfolding** *fca fab* **using** *R2ne* **by** (*simp add*: *field-simps*)
qed

On the circle, $cor\ (r^2) - cnj\ (m - ctr) * (z - ctr)$ factors through $m - z$.

lemma *circle-factor*:
assumes *hm*: *oncircle ctr r m* **and** *pos*: *0* < *r*
shows *complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)*(*z* − *ctr*) = *complex-of-real* (*r*^*2*)*(*m* − *z*)/(*m* − *ctr*)
proof −
have *Mne*: *m* − *ctr* ≠ *0* **using** *oncircle-ne-centre*[*OF hm pos*] **by** *simp*
have *cnjM*: *cnj* (*m* − *ctr*) = *complex-of-real* (*r*^*2*)/(*m* − *ctr*) **using** *cnj-diff-on-circle*[*OF hm pos*] **.**
have *complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)*(*z* − *ctr*) = *complex-of-real* (*r*^*2*) − *complex-of-real* (*r*^*2*)/(*m* − *ctr*)*(*z* − *ctr*)
  by (*subst cnjM*) (*rule refl*)
also have ... = *complex-of-real* (*r*^*2*)*(*m* − *z*)/(*m* − *ctr*) **using** *Mne* **by** (*simp add*: *field-simps*)
finally show *?thesis* **.**
qed

When m too lies on the circle, the two foot differences take a particularly clean form.

lemma *foot-diff-PR-circ*:
assumes *ha*: *oncircle ctr r a* **and** *hb*: *oncircle ctr r b* **and** *hc*: *oncircle ctr r c*
  and *hm*: *oncircle ctr r m* **and** *pos*: *0* < *r* **and** *dab*: *a* ≠ *b* **and** *dbc*: *b* ≠ *c*

shows *foot b c m* − *foot a b m* = (*c* − *a*)∗(*m* − *b*)/(*2*∗(*m* − *ctr*))
proof −
have *R2ne*: *complex-of-real* (*r*^*2*) ≠ *0* **using** *pos* **by** *simp*
have *Mne*: *m* − *ctr* ≠ *0* **using** *oncircle-ne-centre*[*OF hm pos*] **by** *simp*
have *foot b c m* − *foot a b m* = (*c* − *a*)∗(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)∗(*b* − *ctr*))/(*2*∗*complex-of-real* (*r*^*2*))
by (*rule foot-diff-PR*[*OF ha hb hc pos dab dbc*])
also have ... = (*c* − *a*)∗(*complex-of-real* (*r*^*2*)∗(*m* − *b*)/(*m* − *ctr*))/(*2*∗*complex-of-real* (*r*^*2*))
by (*subst circle-factor*[*OF hm pos*]) (*rule refl*)
also have ... = (*c* − *a*)∗(*m* − *b*)/(*2*∗(*m* − *ctr*)) **using** *R2ne Mne* **by** (*simp add*: *field-simps*)
finally show *?thesis* **.**
qed

lemma *foot-diff-QR-circ*:
assumes *ha*: *oncircle ctr r a* **and** *hb*: *oncircle ctr r b* **and** *hc*: *oncircle ctr r c*
and *hm*: *oncircle ctr r m* **and** *pos*: *0* < *r* **and** *dab*: *a* ≠ *b* **and** *dca*: *c* ≠ *a*
shows *foot c a m* − *foot a b m* = (*c* − *b*)∗(*m* − *a*)/(*2*∗(*m* − *ctr*))
proof −
have *R2ne*: *complex-of-real* (*r*^*2*) ≠ *0* **using** *pos* **by** *simp*
have *Mne*: *m* − *ctr* ≠ *0* **using** *oncircle-ne-centre*[*OF hm pos*] **by** *simp*
have *foot c a m* − *foot a b m* = (*c* − *b*)∗(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)∗(*a* − *ctr*))/(*2*∗*complex-of-real* (*r*^*2*))
by (*rule foot-diff-QR*[*OF ha hb hc pos dab dca*])
also have ... = (*c* − *b*)∗(*complex-of-real* (*r*^*2*)∗(*m* − *a*)/(*m* − *ctr*))/(*2*∗*complex-of-real* (*r*^*2*))
by (*subst circle-factor*[*OF hm pos*]) (*rule refl*)
also have ... = (*c* − *b*)∗(*m* − *a*)/(*2*∗(*m* − *ctr*)) **using** *R2ne Mne* **by** (*simp add*: *field-simps*)
finally show *?thesis* **.**
qed

17 The Wallace-Simson line

Forward direction: if m lies on the circumcircle of the nondegenerate triangle a, b, c, then the three feet of the perpendiculars from m to the side lines are collinear. The key step is that the ratio of two foot differences equals a cross ratio of four concyclic points, hence is real.

theorem *simson-line*:
fixes *a b c m ctr* :: *complex* **and** *r* :: *real*
assumes *hncol*: ¬ *collinear a b c*
and *ha*: *oncircle ctr r a* **and** *hb*: *oncircle ctr r b* **and** *hc*: *oncircle ctr r c*
and *hm*: *oncircle ctr r m* **and** *pos*: *0* < *r*
shows *collinear* (*foot b c m*) (*foot c a m*) (*foot a b m*)
proof −
have *dist*: *a* ≠ *b* ∧ *a* ≠ *c* ∧ *b* ≠ *c* **using** *ncol-distinct*[*OF hncol*] **.**
have *dab*: *a* ≠ *b* **and** *dac*: *a* ≠ *c* **and** *dbc*: *b* ≠ *c* **using** *dist* **by** *auto*

```
  have dca: c ≠ a using dac by simp
  have Mne: m − ctr ≠ 0 using oncircle-ne-centre[OF hm pos] by simp
  have hPR: foot b c m − foot a b m = (c − a)∗(m − b)/(2∗(m − ctr))
    by (rule foot-diff-PR-circ[OF ha hb hc hm pos dab dbc])
  have hQR: foot c a m − foot a b m = (c − b)∗(m − a)/(2∗(m − ctr))
    by (rule foot-diff-QR-circ[OF ha hb hc hm pos dab dca])
  show ?thesis
  proof (cases m = a)
    case True
    have foot c a m − foot a b m = 0 using hQR True by simp
    hence foot c a m = foot a b m by simp
    thus ?thesis by (simp add: collinear-refl)
  next
    case False
    hence dma: m ≠ a by simp
    have dcb: c ≠ b using dbc by simp
    have Bne: (c − b)∗(m − a) ≠ 0 using dcb dma by simp
    have D2ne: 2∗(m − ctr) ≠ 0 using Mne by simp
    have RQ: foot a b m ≠ foot c a m
    proof
      assume foot a b m = foot c a m
      hence (c − b)∗(m − a)/(2∗(m − ctr)) = 0 using hQR by simp
      thus False using Bne Mne by simp
    qed
    have ratio: (foot b c m − foot a b m)/(foot c a m − foot a b m) = complex-cross-ratio c m a b
    proof −
      have (foot b c m − foot a b m)/(foot c a m − foot a b m) = ((c − a)∗(m − b)/(2∗(m − ctr)))/((c − b)∗(m − a)/(2∗(m − ctr)))
        using hPR hQR by simp
      also have ... = (c − a)∗(m − b)/((c − b)∗(m − a))
        using D2ne by simp
      also have ... = complex-cross-ratio c m a b
        by (simp add: complex-cross-ratio-def)
      finally show ?thesis .
    qed
    have cr-real: complex-cross-ratio c m a b ∈ ℝ
      by (rule concyclic-complex-cross-ratio-real[OF hc hm ha hb pos dcb dma])
    have (foot b c m − foot a b m)/(foot c a m − foot a b m) ∈ ℝ
      using ratio cr-real by simp
    hence collinear (foot a b m) (foot b c m) (foot c a m)
      using collinear-iff-real-ratio[OF RQ] by simp
    thus ?thesis by (rule collinear-cycle)
  qed
qed
```

18 The converse

A field identity that clears the three circle denominators from the collinearity relation of the feet. It is stated over an arbitrary field so that the algebraic simplification can be discharged uniformly.

lemma *converse-key-gen*:
  fixes *A B C ca cb cc R D cm* :: *'a::field*
  assumes *A0*: *A ≠ 0* **and** *B0*: *B ≠ 0* **and** *C0*: *C ≠ 0*
    and *hA*: *A ∗ ca = R* **and** *hB*: *B ∗ cb = R* **and** *hC*: *C ∗ cc = R*
  shows
    A ∗ B ∗ C ∗
      ((cc − ca) ∗ (R − D ∗ cb) ∗ (C − B) ∗ (R − cm ∗ A) −
       (C − A) ∗ (R − cm ∗ B) ∗ (cc − cb) ∗ (R − D ∗ ca))
     = R^2 ∗ (A − B) ∗ (A − C) ∗ (B − C) ∗ (R − cm ∗ D)
proof −
  have *ca*: *ca = R / A* **using** *hA A0* **by** (*simp add*: *field-simps*)
  have *cb*: *cb = R / B* **using** *hB B0* **by** (*simp add*: *field-simps*)
  have *cc*: *cc = R / C* **using** *hC C0* **by** (*simp add*: *field-simps*)
  show *?thesis* **unfolding** *ca cb cc* **using** *A0 B0 C0* **by** (*simp add*: *field-simps*)
(*simp add*: *algebra-simps power2-eq-square*)
qed

Specialised to the circle, with the conjugate substitutions folded back into differences of the original points.

lemma *converse-key-complex*:
  fixes *a b c m ctr* :: *complex* **and** *r* :: *real*
  assumes *Ane*: *a − ctr ≠ 0* **and** *Bne*: *b − ctr ≠ 0* **and** *Cne*: *c − ctr ≠ 0*
    and *hA*: *(a − ctr)∗cnj (a − ctr) = complex-of-real (r^2)*
    and *hB*: *(b − ctr)∗cnj (b − ctr) = complex-of-real (r^2)*
    and *hC*: *(c − ctr)∗cnj (c − ctr) = complex-of-real (r^2)*
  shows
    (a − ctr) ∗ (b − ctr) ∗ (c − ctr) ∗
      (cnj (c − a) ∗
        (complex-of-real (r^2) − (m − ctr) ∗ cnj (b − ctr)) ∗
        (c − b) ∗
        (complex-of-real (r^2) − cnj (m − ctr) ∗ (a − ctr)) −
       (c − a) ∗
        (complex-of-real (r^2) − cnj (m − ctr) ∗ (b − ctr)) ∗
        cnj (c − b) ∗
        (complex-of-real (r^2) − (m − ctr) ∗ cnj (a − ctr)))
     = (complex-of-real (r^2))^2 ∗ (a − b) ∗ (a − c) ∗ (b − c) ∗
        (complex-of-real (r^2) − cnj (m − ctr) ∗ (m − ctr))
proof −
  have *key*:
    (a − ctr) ∗ (b − ctr) ∗ (c − ctr) ∗
      ((cnj (c − ctr) − cnj (a − ctr)) ∗
        (complex-of-real (r^2) − (m − ctr) ∗ cnj (b − ctr)) ∗
        ((c − ctr) − (b − ctr)) ∗
        (complex-of-real (r^2) − cnj (m − ctr) ∗ (a − ctr)) −

((c − ctr) − (a − ctr)) ∗
(complex-of-real (r^2) − cnj (m − ctr) ∗ (b − ctr)) ∗
(cnj (c − ctr) − cnj (b − ctr)) ∗
(complex-of-real (r^2) − (m − ctr) ∗ cnj (a − ctr)))
= (complex-of-real (r^2))^2 ∗
((a − ctr) − (b − ctr)) ∗
((a − ctr) − (c − ctr)) ∗
((b − ctr) − (c − ctr)) ∗
(complex-of-real (r^2) − cnj (m − ctr) ∗ (m − ctr))
by (rule converse-key-gen[OF Ane Bne Cne hA hB hC])
have g1: cnj(c−ctr)−cnj(a−ctr) = cnj(c−a) **by** (simp add: complex-cnj-diff)
have g4: cnj(c−ctr)−cnj(b−ctr) = cnj(c−b) **by** (simp add: complex-cnj-diff)
have g2: (c−ctr)−(b−ctr) = c−b **by** simp
have g3: (c−ctr)−(a−ctr) = c−a **by** simp
have g5: (a−ctr)−(b−ctr) = a−b **by** simp
have g6: (a−ctr)−(c−ctr) = a−c **by** simp
have g7: (b−ctr)−(c−ctr) = b−c **by** simp
show ?thesis **using** key **by** (simp only: g1 g4 g2 g3 g5 g6 g7)
qed

Converse direction: if the three feet of the perpendiculars from m to the side lines of the nondegenerate triangle are collinear, then m lies on the circumcircle.

theorem simson-line-converse:
fixes a b c m ctr :: complex **and** r :: real
assumes hncol: ¬ collinear a b c
and ha: oncircle ctr r a **and** hb: oncircle ctr r b **and** hc: oncircle ctr r c
and pos: 0 < r
and hcol: collinear (foot b c m) (foot c a m) (foot a b m)
shows oncircle ctr r m
proof −
have dist: a ≠ b ∧ a ≠ c ∧ b ≠ c **using** ncol-distinct[OF hncol] **.**
have dab: a ≠ b **and** dac: a ≠ c **and** dbc: b ≠ c **using** dist **by** auto
have dca: c ≠ a **using** dac **by** simp
have Ane: a − ctr ≠ 0 **using** oncircle-ne-centre[OF ha pos] **by** simp
have Bne: b − ctr ≠ 0 **using** oncircle-ne-centre[OF hb pos] **by** simp
have Cne: c − ctr ≠ 0 **using** oncircle-ne-centre[OF hc pos] **by** simp
have R2ne: complex-of-real (r^2) ≠ 0 **using** pos **by** simp
have hA: (a − ctr)∗cnj (a − ctr) = complex-of-real (r^2) **using** oncircle-norm-sq[OF ha] **.**
have hB: (b − ctr)∗cnj (b − ctr) = complex-of-real (r^2) **using** oncircle-norm-sq[OF hb] **.**
have hC: (c − ctr)∗cnj (c − ctr) = complex-of-real (r^2) **using** oncircle-norm-sq[OF hc] **.**
have hPR: foot b c m − foot a b m = (c − a)∗(complex-of-real (r^2) − cnj (m − ctr)∗(b − ctr))/(2∗complex-of-real (r^2))
by (rule foot-diff-PR[OF ha hb hc pos dab dbc])
have hQR: foot c a m − foot a b m = (c − b)∗(complex-of-real (r^2) − cnj (m − ctr)∗(a − ctr))/(2∗complex-of-real (r^2))

by (*rule foot-diff-QR*[*OF ha hb hc pos dab dca*])
have *col2*: *collinear* (*foot a b m*) (*foot b c m*) (*foot c a m*)
using *collinear-cycle*[*OF collinear-cycle*[*OF hcol*]] **.**
have *Imraw*: *Im* (*cnj* (*foot b c m* − *foot a b m*) ∗ (*foot c a m* − *foot a b m*)) = *0*
using *col2* **by** (*simp add*: *collinear-iff-cross*)
have *cross*: *cnj* (*foot b c m* − *foot a b m*) ∗ (*foot c a m* − *foot a b m*) = (*foot b c m* − *foot a b m*) ∗ *cnj* (*foot c a m* − *foot a b m*)
by (*rule im-zero-cross*[*OF Imraw*])
have *inner0*:
cnj (*c* − *a*) ∗
(*complex-of-real* (*r*^*2*) − (*m* − *ctr*) ∗ *cnj* (*b* − *ctr*)) ∗
(*c* − *b*) ∗
(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*) ∗ (*a* − *ctr*)) −
(*c* − *a*) ∗
(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*) ∗ (*b* − *ctr*)) ∗
cnj (*c* − *b*) ∗
(*complex-of-real* (*r*^*2*) − (*m* − *ctr*) ∗ *cnj* (*a* − *ctr*))
= *0*
using *cross*[*unfolded hPR hQR*] *R2ne* **by** (*simp add*: *complex-cnj-divide complex-cnj-mult complex-cnj-diff field-simps*)
have *keyeq*:
(*a* − *ctr*) ∗ (*b* − *ctr*) ∗ (*c* − *ctr*) ∗
(*cnj* (*c* − *a*) ∗
(*complex-of-real* (*r*^*2*) − (*m* − *ctr*) ∗ *cnj* (*b* − *ctr*)) ∗
(*c* − *b*) ∗
(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*) ∗ (*a* − *ctr*)) −
(*c* − *a*) ∗
(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*) ∗ (*b* − *ctr*)) ∗
cnj (*c* − *b*) ∗
(*complex-of-real* (*r*^*2*) − (*m* − *ctr*) ∗ *cnj* (*a* − *ctr*)))
= (*complex-of-real* (*r*^*2*))^*2* ∗ (*a* − *b*) ∗ (*a* − *c*) ∗ (*b* − *c*) ∗
(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*) ∗ (*m* − *ctr*))
by (*rule converse-key-complex*[*OF Ane Bne Cne hA hB hC*])
have *RHS0*: (*complex-of-real* (*r*^*2*))^*2*∗(*a* − *b*)∗(*a* − *c*)∗(*b* − *c*)∗(*complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)∗(*m* − *ctr*)) = *0*
using *keyeq inner0* **by** *simp*
have *abcne*: (*a* − *b*)∗(*a* − *c*)∗(*b* − *c*) ≠ *0* **using** *dab dac dbc* **by** *simp*
have *zero*: *complex-of-real* (*r*^*2*) − *cnj* (*m* − *ctr*)∗(*m* − *ctr*) = *0*
using *RHS0 R2ne abcne* **by** *simp*
have (*m* − *ctr*)∗*cnj* (*m* − *ctr*) = *complex-of-real* (*r*^*2*)
using *zero* **by** (*simp add*: *mult.commute*)
thus *oncircle ctr r m* **by** (*rule oncircle-of-norm-sq*[*OF - pos*])
qed

The two directions combine into the usual characterisation of the circumcircle by collinearity of the three perpendicular feet.

theorem *simson-line-iff*:
fixes *a b c m ctr* :: *complex* **and** *r* :: *real*
assumes *hncol*: ¬ *collinear a b c*

```
    and ha: oncircle ctr r a and hb: oncircle ctr r b and hc: oncircle ctr r c
    and pos: 0 < r
  shows
    oncircle ctr r m ⟷
      collinear (foot b c m) (foot c a m) (foot a b m)
proof
  assume hm: oncircle ctr r m
  show collinear (foot b c m) (foot c a m) (foot a b m)
    by (rule simson-line[OF hncol ha hb hc hm pos])
next
  assume hcol: collinear (foot b c m) (foot c a m) (foot a b m)
  show oncircle ctr r m
    by (rule simson-line-converse[OF hncol ha hb hc pos hcol])
qed

end
```

References

[1] H. S. M. Coxeter and S. L. Greitzer. *Geometry Revisited*, volume 19 of *New Mathematical Library*. Mathematical Association of America, 1967.

[2] R. A. Johnson. *Modern Geometry: An Elementary Treatise on the Geometry of the Triangle and the Circle*. Houghton Mifflin, Boston, 1929.

[3] F. Marić and D. Simić. Complex geometry. *Archive of Formal Proofs*, 2019. https://isa-afp.org/entries/Complex_Geometry.html, Formal proof development.

[4] W. Wallace. Some geometrical porisms, with examples of their application to the solution of problems. *Transactions of the Royal Society of Edinburgh*, 4:107–134, 1798. The line of the feet of the perpendiculars is due to Wallace and is commonly misattributed to Robert Simson; the publication is sometimes cited as 1799.